
	Reason for change:
	1) The MCC implementation of R5s170028 brought some issues with IMS. All TCs (13.4.3.2, 13.4.3.14, 13.4.3.16) need to handle the optional Deregistration and PDP context deactivation procedures that can happen both, one or none of them.

Some problems have been observed if the deregistration happens at the end of RAU procedure. If the trigger to f_IMS_IpcanReleaseWithOptionalDeregistrationOnAnyPTC is sent too early after RAU complete, there could be unexpected REGISTER message received and the TC might fail.

Including wait time of 2 seconds after RAU complete can provide enough time for UE to finish up deregistration.

2)  Same change should be applied also to UTRAN part of TC 13.4.3.18.

This document sums up all the changes on top of R5s170028_MCC160Comments.

#draftProseCR would be raised for next RAN5 meeting.

Change 1
	Function name
	f_IMS_IpcanReleaseWithOptionalDeregistrationOnAnyPTC

	Reason for change
	The solution implemented by MCC in Change 1 of R5s170028 does not consider several TCs that require to have p_WaitForDeRegistration set to false in postamble (i.e. TC 8.3, 11.1, 19.5.9).
Proposed is to use the original Change 1 implementation from R5s170028.
Additional change exits the while loop when trigger is received and no Deregistration happened.

#draftProseCR would be raised for next RAN5 meeting

	Summary of change
	· Reverted to original change 1 of R5s170028.
· Added v_WaitForDeRegistration in altsep to exit loop also if no deregistration happened

	TTCN module
	Common\IMS\IMS_Procedures_Registration.ttcn

	MCC160 Comment
	


Before change (original WK49)
  function f_IMS_IpcanReleaseWithOptionalDeregistrationOnAnyPTC(IMS_IPCAN_CO_ORD_PORT p_Port,

                                                                boolean p_WaitForTrigger := true,

                                                                boolean p_AllowRetransmission := false,

                                                                boolean p_WaitForDeRegistration := true) runs on IMS_PTC return boolean

  { /* @sic R5s140710 - MCC160 Implementation: parameter p_AllowRetransmission sic@ */

    var IMS_DATA_REQ v_IMS_REGISTER_REQ_ByRef;

    var template (present) SipUrl v_SipUrl := cr_SipUri_HostPort(f_IMS_PTC_ImsInfo_GetHomeDomainName());

    var InternetProtocol_Type v_ProtocolForRequestInDL := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var boolean v_WaitForDeRegistration := p_WaitForDeRegistration;

    var boolean v_WaitForTrigger := p_WaitForTrigger;

    var boolean v_WaitForUnsubscribe_RegEvent   := f_IMS_PTC_ImsInfo_DialogIsStarted(regEventDialog);   /* @sic R5-153756 sic@ */

    var boolean v_WaitForUnsubscribe_MsgSummary := f_IMS_PTC_ImsInfo_DialogIsStarted(msgSummaryDialog); /* @sic R5-153756 sic@ */

    var boolean v_WaitForUnsubscribe_ConfEvent  := f_IMS_PTC_ImsInfo_DialogIsStarted(confEventDialog);  /* @sic R5-153756 sic@ */

    var template (value) MessageHeader v_MessageHeader_Notify_RegEvent_ByRef;

    var template (value) MessageHeader v_MessageHeader_Notify_MsgSummary_ByRef;

    var template (value) MessageHeader v_MessageHeader_Notify_ConfEvent_ByRef;

    var boolean v_WaitFor200OK_RegEvent := false;

    var boolean v_WaitFor200OK_MsgSummary := false;

    var boolean v_WaitFor200OK_ConfEvent := false;

    var default v_RetransmissionHandler := null;

    while (v_WaitForDeRegistration or v_WaitFor200OK_RegEvent or v_WaitFor200OK_MsgSummary or v_WaitFor200OK_ConfEvent or v_WaitForTrigger) { // @sic R5-141113: de-registration is not mandatory sic@

      alt {

        [v_WaitForDeRegistration and v_WaitForUnsubscribe_RegEvent] a_IMS_Deregister_Unsubcribe(regEventDialog, v_ProtocolForRequestInDL, v_MessageHeader_Notify_RegEvent_ByRef) {  /* @sic R5-153756 sic@ */

          v_WaitForUnsubscribe_RegEvent := false;

          v_WaitFor200OK_RegEvent := true;

        }

        [v_WaitForDeRegistration and v_WaitForUnsubscribe_ConfEvent] a_IMS_Deregister_Unsubcribe(confEventDialog, v_ProtocolForRequestInDL, v_MessageHeader_Notify_ConfEvent_ByRef) {  /* @sic R5-153756 sic@ */

          v_WaitForUnsubscribe_ConfEvent := false;

          v_WaitFor200OK_ConfEvent := true;

        }

        [v_WaitForDeRegistration and v_WaitForUnsubscribe_MsgSummary] a_IMS_Deregister_Unsubcribe(msgSummaryDialog, v_ProtocolForRequestInDL, v_MessageHeader_Notify_MsgSummary_ByRef) {  /* @sic R5-153756 sic@ */

          v_WaitForUnsubscribe_MsgSummary := false;

          v_WaitFor200OK_MsgSummary := true;

        }

        [v_WaitForDeRegistration] a_IMS_ReceiveRequest(car_IMS_Register_Request(cr_REGISTER_Request(v_SipUrl)), v_IMS_REGISTER_REQ_ByRef) {

          if (p_AllowRetransmission) {

            v_RetransmissionHandler := activate(a_IMS_RetransmissionHandler(v_IMS_REGISTER_REQ_ByRef));  /* when re-transmissions are allowed they may colide with f_Delay in f_IMS_TCP_CloseHandling

                                                                                                            or with the confirmation in f_IMS_CloseTCP (if any); */

          }

          f_IMS_PTC_Deregistration(v_IMS_REGISTER_REQ_ByRef);

          v_WaitForDeRegistration := false;

        }

        [v_WaitFor200OK_RegEvent] a_IMS_ReceiveResponse200OK(v_MessageHeader_Notify_RegEvent_ByRef, v_ProtocolForRequestInDL) {  // @sic R5-168950 sic@

          v_WaitFor200OK_RegEvent := false;

        }

        [v_WaitFor200OK_ConfEvent] a_IMS_ReceiveResponse200OK(v_MessageHeader_Notify_ConfEvent_ByRef, v_ProtocolForRequestInDL) {  // @sic R5-168950 sic@

          v_WaitFor200OK_ConfEvent := false;

        }

        [v_WaitFor200OK_MsgSummary] a_IMS_ReceiveResponse200OK(v_MessageHeader_Notify_MsgSummary_ByRef, v_ProtocolForRequestInDL) {  // @sic R5-168950 sic@

          v_WaitFor200OK_MsgSummary := false;

        }

        [v_WaitForTrigger] p_Port.receive(cmr_IMS_IPCAN_Trigger) { // @sic R5-142919 sic@

          v_WaitForTrigger := false;

          if (v_WaitFor200OK_RegEvent or v_WaitFor200OK_MsgSummary or v_WaitFor200OK_ConfEvent) {  // @sic R5-168950 sic@

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "still waiting for 200 OK for NOTIFY");

          }

        }

      }

    }

    deactivate(v_RetransmissionHandler);

    return not v_WaitForDeRegistration; // @sic R5-154047 sic@

  }
After change

function f_IMS_IpcanReleaseWithOptionalDeregistrationOnAnyPTC(IMS_IPCAN_CO_ORD_PORT p_Port,

                                                                boolean p_WaitForTrigger := true,

                                                                boolean p_AllowRetransmission := false,

                                                                boolean p_WaitForDeRegistration := true) runs on IMS_PTC return boolean

  { /* @sic R5s140710 - MCC160 Implementation: parameter p_AllowRetransmission sic@ */

    var IMS_DATA_REQ v_IMS_REGISTER_REQ_ByRef;

    var template (present) SipUrl v_SipUrl := cr_SipUri_HostPort(f_IMS_PTC_ImsInfo_GetHomeDomainName());

    var InternetProtocol_Type v_ProtocolForRequestInDL := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var boolean v_WaitForDeRegistration := p_WaitForDeRegistration;

    var boolean v_WaitForTrigger := p_WaitForTrigger;

    var boolean v_WaitForUnsubscribe_RegEvent   := f_IMS_PTC_ImsInfo_DialogIsStarted(regEventDialog);   /* @sic R5-153756 sic@ */

    var boolean v_WaitForUnsubscribe_MsgSummary := f_IMS_PTC_ImsInfo_DialogIsStarted(msgSummaryDialog); /* @sic R5-153756 sic@ */

    var boolean v_WaitForUnsubscribe_ConfEvent  := f_IMS_PTC_ImsInfo_DialogIsStarted(confEventDialog);  /* @sic R5-153756 sic@ */

    var template (value) MessageHeader v_MessageHeader_Notify_RegEvent_ByRef;

    var template (value) MessageHeader v_MessageHeader_Notify_MsgSummary_ByRef;

    var template (value) MessageHeader v_MessageHeader_Notify_ConfEvent_ByRef;

    var boolean v_WaitFor200OK_RegEvent := false;

    var boolean v_WaitFor200OK_MsgSummary := false;

    var boolean v_WaitFor200OK_ConfEvent := false;

    var boolean v_DeRegistered := false;//R5s170028 Change 1
    var default v_RetransmissionHandler := null;

    while (v_WaitForDeRegistration or v_WaitFor200OK_RegEvent or v_WaitFor200OK_MsgSummary or v_WaitFor200OK_ConfEvent or v_WaitForTrigger)  { // @sic R5-141113: de-registration is not mandatory sic@

      alt {

        [v_WaitForDeRegistration and v_WaitForUnsubscribe_RegEvent] a_IMS_Deregister_Unsubcribe(regEventDialog, v_ProtocolForRequestInDL, v_MessageHeader_Notify_RegEvent_ByRef) {  /* @sic R5-153756 sic@ */

          v_WaitForUnsubscribe_RegEvent := false;

          v_WaitFor200OK_RegEvent := true;

        }

        [v_WaitForDeRegistration and v_WaitForUnsubscribe_ConfEvent] a_IMS_Deregister_Unsubcribe(confEventDialog, v_ProtocolForRequestInDL, v_MessageHeader_Notify_ConfEvent_ByRef) {  /* @sic R5-153756 sic@ */

          v_WaitForUnsubscribe_ConfEvent := false;

          v_WaitFor200OK_ConfEvent := true;

        }

        [v_WaitForDeRegistration and v_WaitForUnsubscribe_MsgSummary] a_IMS_Deregister_Unsubcribe(msgSummaryDialog, v_ProtocolForRequestInDL, v_MessageHeader_Notify_MsgSummary_ByRef) {  /* @sic R5-153756 sic@ */

          v_WaitForUnsubscribe_MsgSummary := false;

          v_WaitFor200OK_MsgSummary := true;

        }

        [v_WaitForDeRegistration] a_IMS_ReceiveRequest(car_IMS_Register_Request(cr_REGISTER_Request(v_SipUrl)), v_IMS_REGISTER_REQ_ByRef) {

          if (p_AllowRetransmission) {

            v_RetransmissionHandler := activate(a_IMS_RetransmissionHandler(v_IMS_REGISTER_REQ_ByRef));  /* when re-transmissions are allowed they may colide with f_Delay in f_IMS_TCP_CloseHandling

                                                                                                            or with the confirmation in f_IMS_CloseTCP (if any); */

          }

          f_IMS_PTC_Deregistration(v_IMS_REGISTER_REQ_ByRef);

          v_WaitForDeRegistration := false;

          v_DeRegistered := true; //R5s170028 Change 1
        }

        [v_WaitFor200OK_RegEvent] a_IMS_ReceiveResponse200OK(v_MessageHeader_Notify_RegEvent_ByRef, v_ProtocolForRequestInDL) {  // @sic R5-168950 sic@

          v_WaitFor200OK_RegEvent := false;

        }

        [v_WaitFor200OK_ConfEvent] a_IMS_ReceiveResponse200OK(v_MessageHeader_Notify_ConfEvent_ByRef, v_ProtocolForRequestInDL) {  // @sic R5-168950 sic@

          v_WaitFor200OK_ConfEvent := false;

        }

        [v_WaitFor200OK_MsgSummary] a_IMS_ReceiveResponse200OK(v_MessageHeader_Notify_MsgSummary_ByRef, v_ProtocolForRequestInDL) {  // @sic R5-168950 sic@

          v_WaitFor200OK_MsgSummary := false;

        }

        [v_WaitForTrigger] p_Port.receive(cmr_IMS_IPCAN_Trigger) { // @sic R5-142919 sic@

          v_WaitForTrigger := false;

          v_WaitForDeRegistration := false;
          if (v_WaitFor200OK_RegEvent or v_WaitFor200OK_MsgSummary or v_WaitFor200OK_ConfEvent) {  // @sic R5-168950 sic@

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "still waiting for 200 OK for NOTIFY");

          }

        }

      }

    }

    deactivate(v_RetransmissionHandler);

    return v_DeRegistered; //R5s170028 Change 1   // @sic R5-154047 sic@

  }
Change 2
	Function name
	a_IMS_C24_CheckInvite

	Reason for change
	Not needed change from MCC160 Implementation in Step 4 of R5s170028 reverted back to original wk49 implemenentation. Changed implementation in test body of the TC 13.4.3.2 -> Change 3.

	Summary of change
	Removed MCC implementation in step 4 of R5s170028.

	TTCN module
	\Common\IMS\IMS_CommonSRVCC.ttcn

	MCC160 Comment
	


After change (original WK49 impl.)
  altstep a_IMS_C24_CheckInvite() runs on IMS_PTC

  {

    [] IMS_Server.check(receive(car_IMS_Invite_Request())) {} // no repeat because need to drop out of calling alt statement

    [] any timer.timeout {} // no repeat because need to drop out of calling alt statement

  }
Change 3
	Function name
	f_TC_13_4_3_2_UTRAN

	Reason for change
	MCC implementation in Change 4 of R5s170028 is missing f_IMS_IPCAN_SendCoOrdMsg after f_UTRAN_SRVCC_WithHandover.
Added delay of 2 second before trigger to IPCAN to provide UE with enough time to proceed with optional deregistration after RAU complete message is received.

Activated default to process the otional PDP context deactivation request/accept after  RAU procedure. 

Dismantled altstep as cmr_IMS_IPCAN_Abort is not needed anymore.

	Summary of change
	· Added trigger to IPCAN.
· Added 2s delay.

· Activated default behaviour

· Dismantled altstep

	TTCN module
	\LTE_IRAT\13\SRVCC_EUTRA_UTRAN.ttcn

	MCC160 Comment
	


Before change (R5s170028 MCC160comments)
    function f_TC_13_4_3_2_UTRAN() runs on UTRAN_IRAT_PTC

  { /* Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / SRVCC */

    //@sic R5-120667 sic@

    var integer v_T0_CPICH_Cell5;

    var integer v_T1_CPICH_Cell5;

    var integer v_T0_PCCPCH_Cell5;

    var integer v_T1_PCCPCH_Cell5;

    v_T0_CPICH_Cell5 := -88;

    v_T1_CPICH_Cell5 := -64;//@sic R5-125742 sic@

    v_T0_PCCPCH_Cell5 := -88;

    v_T1_PCCPCH_Cell5 := -64;//@sic R5-125742 sic@

….

    //according to 36.508 cl. 6.4.3.9.5,

    f_UTRAN_SRVCC_WithHandover(utran_Cell5, cs_PDP_ContextStatusNSAPI5, true, '0'B); //@sic R5s140096, R5-144667, R5-150397 sic@

    alt { // @sic R5s170028 sic@

      [] a_UTRAN_DeactivatePDPContext_MO () {

        repeat;

        }

      [] IMS[tsc_Index_PDN1].check (receive(cmr_IMS_IPCAN_Trigger)){

        // Ok, we can stop waiting now

        }

      [] IMS[tsc_Index_PDN1].receive (cmr_IMS_IPCAN_Abort){

        f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); 

        f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]); // @sic R5-144667 sic@
        // Stop waiting for PDP Context Deactivate now

        }

    }    

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell to EUTRA PTC Test case is now finished on E-UTRA side

    f_UTRAN_TestBody_Set(false);

    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U5_CSCALL,-,cs_TI_MT);

  }
After change

    function f_TC_13_4_3_2_UTRAN() runs on UTRAN_IRAT_PTC

  { /* Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / SRVCC */

    //@sic R5-120667 sic@

    var integer v_T0_CPICH_Cell5;

    var integer v_T1_CPICH_Cell5;

    var integer v_T0_PCCPCH_Cell5;

    var integer v_T1_PCCPCH_Cell5;

    v_T0_CPICH_Cell5 := -88;

    v_T1_CPICH_Cell5 := -64;//@sic R5-125742 sic@

    v_T0_PCCPCH_Cell5 := -88;

    v_T1_PCCPCH_Cell5 := -64;//@sic R5-125742 sic@

    var default v_DefaultRef;
...

    //according to 36.508 cl. 6.4.3.9.5,

    f_UTRAN_SRVCC_WithHandover(utran_Cell5, cs_PDP_ContextStatusNSAPI5, true, '0'B); //@sic R5s140096, R5-144667, R5-150397 sic@

    v_DefaultRef := activate (a_UTRAN_DeactivatePDPContext_MO());
    f_Delay(2.0);

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // new optional deregistration waits for a trigger @sic R5-154047 sic@
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]); // @sic R5-144667 sic@

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell to EUTRA PTC Test case is now finished on E-UTRA side

    f_UTRAN_TestBody_Set(false);

    deactivate (v_DefaultRef);
    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U5_CSCALL,-,cs_TI_MT);

  }
Change 4
	Function name
	f_TC_13_4_3_14_UTRAN

	Reason for change
	 Added delay of 2 second before sending CONNECT message to provide UE with enough time to proceed with optional deregistration after RAU complete message is received. Trigger to IPCAN is sent right after CONNECT ACK.

Activated default to process the optional PDP context deactivation request/accept after  RAU procedure. 

Dismantled altstep as cmr_IMS_IPCAN_Abort is not needed anymore.
#draftProseCR would be raised for next RAN5 meeting

	Summary of change
	· Added 2s delay.

· Activated default behaviour.

· Dismantled altstep.

	TTCN module
	\LTE_A_IRAT\13\ASRVCC_EUTRA_UTRAN.ttcn

	MCC160 Comment
	


Before change

       function f_TC_13_4_3_14_UTRAN() runs on UTRAN_IRAT_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MO call

    var integer v_T0_CPICH_Cell5 := -88;

    var integer v_T1_CPICH_Cell5 := -64;

    var integer v_T0_PCCPCH_Cell5 := -88;

    var integer v_T1_PCCPCH_Cell5 := -64;

    var TI v_TI_S;

    var TI v_TI_R;

    f_UTRAN_Init(EUTRA_UTRAN);

    // Create and configure Cell 5

    // Cell started switched on with default power level, -60 and max attenuation

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    // Start sending System Information

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    // Send relevant UTRAN cell info to EUTRA PTC

    f_UTRAN_Send_IR_SysInfoToEUTRA(utran_Cell5);

    f_UTRAN_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //Configure a dedicated channel in the UTRAN FDD with configuration

    // according to 36.108 Table 4.8.3-1 condition UTRA Speech

    f_SS_Config_ConvSpeech_64kPS_RAB(utran_Cell5);

    // wait for trigger from EUTRAN side

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    // Set Cell Power level as per T0

    f_UTRAN_SetCellPower(utran_Cell5, v_T0_CPICH_Cell5, v_T0_PCCPCH_Cell5);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 18" siclog@

    // Set Cell Power level as per T1

    f_UTRAN_SetCellPower(utran_Cell5, v_T1_CPICH_Cell5, v_T1_PCCPCH_Cell5);

    // UE starts registered in EUTRA, wait its Auth Params

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // Set up the ciphering ready for when the UE hands over

    f_UTRAN_ConfigureCipheringBeforeHO_SRVCC(utran_Cell5);

    f_UTRAN_IP_Handling_Change_Configuration (utran_CellDedicated,  cell_DCH_64kPS_RAB_SRB); // @sic R5s140572 sic@

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell SS to transmit a MobilityFromEUTRACommand message

    f_UTRAN_SRVCC_WithHandover(utran_Cell5, cs_PDP_ContextStatusNSAPI5, true, '0'B); // @sic R5-144745, R5-150398 sic@

    v_TI_S.tiVal  := '000'B;

    v_TI_S.tiFlag := '0'B;

    //@siclog "Step 30" siclog@

    U_Dc.send (cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_Connect(v_TI_S)));

    v_TI_R := v_TI_S;

    v_TI_R.tiFlag := '1'B;

    alt { // @sic R5-162791 sic@

      [] U_Dc.receive (car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_ConnAck(v_TI_R))) {
      //@siclog "Step 31" siclog@

      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 31");

      f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // new optional deregistration waits for a trigger @sic R5-154047 sic@

      f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // @sic R5-144745 sic@

      repeat;

      }

      [] a_UTRAN_DeactivatePDPContext_MO () {

        repeat;

      }

      [] IMS[tsc_Index_PDN1].check (receive(cmr_IMS_IPCAN_Trigger)){

      // Ok, we can stop waiting now

      }

      [] IMS[tsc_Index_PDN1].receive (cmr_IMS_IPCAN_Abort){

      // Stop waiting for PDP Context Deactivate now

      }

    }
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell to EUTRA PTC Test case is now finished on E-UTRA side

    f_UTRAN_TestBody_Set(false);

    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U5_CSCALL, -, v_TI_S);

  }
After change

     function f_TC_13_4_3_14_UTRAN() runs on UTRAN_IRAT_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MO call

    var integer v_T0_CPICH_Cell5 := -88;

    var integer v_T1_CPICH_Cell5 := -64;

    var integer v_T0_PCCPCH_Cell5 := -88;

    var integer v_T1_PCCPCH_Cell5 := -64;

    var TI v_TI_S;

    var TI v_TI_R;

    var default v_DefaultRef;
    f_UTRAN_Init(EUTRA_UTRAN);

    // Create and configure Cell 5

    // Cell started switched on with default power level, -60 and max attenuation

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    // Start sending System Information

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    // Send relevant UTRAN cell info to EUTRA PTC

    f_UTRAN_Send_IR_SysInfoToEUTRA(utran_Cell5);

    f_UTRAN_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //Configure a dedicated channel in the UTRAN FDD with configuration

    // according to 36.108 Table 4.8.3-1 condition UTRA Speech

    f_SS_Config_ConvSpeech_64kPS_RAB(utran_Cell5);

    // wait for trigger from EUTRAN side

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    // Set Cell Power level as per T0

    f_UTRAN_SetCellPower(utran_Cell5, v_T0_CPICH_Cell5, v_T0_PCCPCH_Cell5);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 18" siclog@

    // Set Cell Power level as per T1

    f_UTRAN_SetCellPower(utran_Cell5, v_T1_CPICH_Cell5, v_T1_PCCPCH_Cell5);

    // UE starts registered in EUTRA, wait its Auth Params

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // Set up the ciphering ready for when the UE hands over

    f_UTRAN_ConfigureCipheringBeforeHO_SRVCC(utran_Cell5);

    f_UTRAN_IP_Handling_Change_Configuration (utran_CellDedicated,  cell_DCH_64kPS_RAB_SRB); // @sic R5s140572 sic@

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell SS to transmit a MobilityFromEUTRACommand message

    f_UTRAN_SRVCC_WithHandover(utran_Cell5, cs_PDP_ContextStatusNSAPI5, true, '0'B); // @sic R5-144745, R5-150398 sic@

    v_DefaultRef := activate (a_UTRAN_DeactivatePDPContext_MO());

    f_Delay(2.0);
    v_TI_S.tiVal  := '000'B;

    v_TI_S.tiFlag := '0'B;

    //@siclog "Step 30" siclog@

    U_Dc.send (cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_Connect(v_TI_S)));

    v_TI_R := v_TI_S;

    v_TI_R.tiFlag := '1'B;

    U_Dc.receive (car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_ConnAck(v_TI_R))); 

    //@siclog "Step 31" siclog@

    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 31");

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // new optional deregistration waits for a trigger @sic R5-154047 sic@

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // @sic R5-144745 sic@

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);

    deactivate (v_DefaultRef);
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell to EUTRA PTC Test case is now finished on E-UTRA side

    f_UTRAN_TestBody_Set(false);

    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U5_CSCALL, -, v_TI_S);

  }
Change 5
	Function name
	f_TC_13_4_3_14_IMS1

	Reason for change
	Removed the trigger to UTRAN as it is not neede anymore as PDP context deativation is moved to default behaviour. (Altstep dismantled).

	Summary of change
	Removed trigger to UTRAN PTC.

	TTCN module
	\LTE_A_IRAT\13\ASRVCC_EUTRA_IMS.ttcn

	MCC160 Comment
	


Before change

function f_TC_13_4_3_14_IMS1 ()runs on IMS_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MO call

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var MessageHeader v_MessageHeader_Response183;

    var boolean v_DeregisteredEarly := false;

    f_IMS_PTC_Init();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);  // @sic R5s140587 sic@

    f_IMS_Registration();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);  // @sic R5s140587 sic@

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    f_UT_RequestIMSCall(MMI, px_IMS_CalleeUri); // @sic R5-134290 sic@

    /* Table 13.4.3.7.3.2-3: Parallel behaviour

       Steps 5-11 expected sequence defined in annex C.21 of TS 34.229-1 [35].

       NOTE: IMS MO speech call is in alerting phase.

    */

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4();

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    v_MessageHeader_Response183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN); // @sic R5s140572 sic@

    /* TS 36.508 4.5A.10.3-1, parallel behaviour

     *  Steps 5 to 11 expected sequence defined in annex C.21

     */

    fl_IMS_MOCallSetup_AnnexC21_Steps5To11(v_InviteRequestWithSdp, v_MessageHeader_Response183);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_PTC_SetRanType(UTRAN_FDD);

    v_DeregisteredEarly := f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5-154047 sic@

    if (not v_DeregisteredEarly) {  // tell UTRAN to stop waiting for Deactivate PDP context @sic R5-162791 sic@

      f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN, cms_IMS_IPCAN_Abort);

    }
    f_IMS_IPCAN_WaitForTrigger(OtherIPCAN);
    if (not v_DeregisteredEarly) {  // @sic R5-154047 sic@

      f_IMS_SRVCC_RemoveEarlyDialogMO_AnnexC34(v_InviteRequestWithSdp); // @sic R5-144745 sic@

      f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

      v_DeregisteredEarly := f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5-142919 sic@

    } else {

      f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);  // Still need to tell UTRAN to close down

      f_IMS_IPCAN_WaitForTrigger(OtherIPCAN); // Already done deregistration, so trigger sent during UTRAN postamble needs to be swallowed here

    }

  }
After change

  function f_TC_13_4_3_14_IMS1 ()runs on IMS_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MO call

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var MessageHeader v_MessageHeader_Response183;

    var boolean v_DeregisteredEarly := false;

    f_IMS_PTC_Init();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);  // @sic R5s140587 sic@

    f_IMS_Registration();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);  // @sic R5s140587 sic@

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    f_UT_RequestIMSCall(MMI, px_IMS_CalleeUri); // @sic R5-134290 sic@

    /* Table 13.4.3.7.3.2-3: Parallel behaviour

       Steps 5-11 expected sequence defined in annex C.21 of TS 34.229-1 [35].

       NOTE: IMS MO speech call is in alerting phase.

    */

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4();

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    v_MessageHeader_Response183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN); // @sic R5s140572 sic@

    /* TS 36.508 4.5A.10.3-1, parallel behaviour

     *  Steps 5 to 11 expected sequence defined in annex C.21

     */

    fl_IMS_MOCallSetup_AnnexC21_Steps5To11(v_InviteRequestWithSdp, v_MessageHeader_Response183);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_PTC_SetRanType(UTRAN_FDD);
    v_DeregisteredEarly := f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5-154047 sic@

    f_IMS_IPCAN_WaitForTrigger(OtherIPCAN);

    if (not v_DeregisteredEarly) {  // @sic R5-154047 sic@

      f_IMS_SRVCC_RemoveEarlyDialogMO_AnnexC34(v_InviteRequestWithSdp); // @sic R5-144745 sic@

      f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

      v_DeregisteredEarly := f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5-142919 sic@

    } else {

      f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);  // Still need to tell UTRAN to close down

      f_IMS_IPCAN_WaitForTrigger(OtherIPCAN); // Already done deregistration, so trigger sent during UTRAN postamble needs to be swallowed here

    }

  }
Change 6
	Function name
	f_TC_13_4_3_16_UTRAN

	Reason for change
	Added delay of 2 second before sending UT command to answer the call to provide UE with enough time to proceed with optional deregistration after RAU complete message is received. Trigger to IPCAN is sent right after CONNECT ACK.

Activated default to process the optional PDP context deactivation request/accept after  RAU procedure. 

Dismantled altstep as cmr_IMS_IPCAN_Abort is not needed anymore.
#draftProseCR would be raised for next RAN5 meeting

	Summary of change
	· Added 2s delay.

· Activated default behaviour.

· Dismantled altstep.

	TTCN module
	\LTE_A_IRAT\13\ASRVCC_EUTRA_UTRAN.ttcn

	MCC160 Comment
	


Before change

  function f_TC_13_4_3_16_UTRAN() runs on UTRAN_IRAT_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MT call

    var integer v_T0_CPICH_Cell5 := -88;

    var integer v_T1_CPICH_Cell5 := -64;

    var integer v_T0_PCCPCH_Cell5 := -88;

    var integer v_T1_PCCPCH_Cell5 := -64;

    var TI v_TI_S;

    var TI v_TI_R;

    f_UTRAN_Init(EUTRA_UTRAN);

    // Create and configure Cell 5

    // Cell started switched on with default power level, -60 and max attenuation

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    // Start sending System Information

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    // Send relevant UTRAN cell info to EUTRA PTC

    f_UTRAN_Send_IR_SysInfoToEUTRA(utran_Cell5);

    f_UTRAN_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //Configure a dedicated channel in the UTRAN FDD with configuration

    // according to 34.108 cl. 6.10.2.4.1.26

    f_SS_Config_ConvSpeech_64kPS_RAB(utran_Cell5);

    // wait for trigger from EUTRAN side

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    // Set Cell Power level as per T0

    f_UTRAN_SetCellPower(utran_Cell5, v_T0_CPICH_Cell5, v_T0_PCCPCH_Cell5);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 26" siclog@

    // Set Cell Power level as per T1

    f_UTRAN_SetCellPower(utran_Cell5, v_T1_CPICH_Cell5, v_T1_PCCPCH_Cell5);

    // UE starts registered in EUTRA, wait its Auth Params

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // Set up the ciphering ready for when the UE hands over

    f_UTRAN_ConfigureCipheringBeforeHO_SRVCC(utran_Cell5);

    f_UTRAN_IP_Handling_Change_Configuration (utran_CellDedicated,  cell_DCH_64kPS_RAB_SRB); // @sic R5s140572 sic@

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell SS to transmit a MobilityFromEUTRACommand message

    // @siclog "Steps 32 to 37" siclog@

    //according to 36.508 cl. 6.4.3.9.5,

    f_UTRAN_SRVCC_WithHandover(utran_Cell5, cs_PDP_ContextStatusNSAPI5, true, '0'B); // @sic R5-144670, R5-150398 sic@

    f_UT_AnswerCall ( UT );

    v_TI_R.tiVal  := '000'B;

    v_TI_R.tiFlag := '1'B; //@sic R5s140578 sic@

    alt { // @sic R5-162791 sic@

      [] U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_Connect(v_TI_R))) {
        //@siclog "Step 38" siclog@

        v_TI_S := v_TI_R;

        v_TI_S.tiFlag := '0'B;//@sic R5s140578 sic@

        //@siclog "Step 39" siclog@

        U_Dc.send(cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_ConnectAck(v_TI_S)));

        f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // new optional deregistration waits for a trigger @sic R5-154047 sic@

        f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // @sic R5-144670 sic@
        repeat;

        }

      [] a_UTRAN_DeactivatePDPContext_MO () {

        repeat;

        }

      [] IMS[tsc_Index_PDN1].check (receive(cmr_IMS_IPCAN_Trigger)){

        // Ok, we can stop waiting now

        }

      [] IMS[tsc_Index_PDN1].receive (cmr_IMS_IPCAN_Abort){

        // Stop waiting for PDP Context Deactivate now

        }

    }
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell to EUTRA PTC Test case is now finished on E-UTRA side

    f_UTRAN_TestBody_Set(false);

    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U5_CSCALL, -, v_TI_S);

  }
After change

  function f_TC_13_4_3_16_UTRAN() runs on UTRAN_IRAT_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MT call

    var integer v_T0_CPICH_Cell5 := -88;

    var integer v_T1_CPICH_Cell5 := -64;

    var integer v_T0_PCCPCH_Cell5 := -88;

    var integer v_T1_PCCPCH_Cell5 := -64;

    var TI v_TI_S;

    var TI v_TI_R;

    var default v_DefaultRef;
    f_UTRAN_Init(EUTRA_UTRAN);

    // Create and configure Cell 5

    // Cell started switched on with default power level, -60 and max attenuation

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    // Start sending System Information

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    // Send relevant UTRAN cell info to EUTRA PTC

    f_UTRAN_Send_IR_SysInfoToEUTRA(utran_Cell5);

    f_UTRAN_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //Configure a dedicated channel in the UTRAN FDD with configuration

    // according to 34.108 cl. 6.10.2.4.1.26

    f_SS_Config_ConvSpeech_64kPS_RAB(utran_Cell5);

    // wait for trigger from EUTRAN side

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    // Set Cell Power level as per T0

    f_UTRAN_SetCellPower(utran_Cell5, v_T0_CPICH_Cell5, v_T0_PCCPCH_Cell5);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 26" siclog@

    // Set Cell Power level as per T1

    f_UTRAN_SetCellPower(utran_Cell5, v_T1_CPICH_Cell5, v_T1_PCCPCH_Cell5);

    // UE starts registered in EUTRA, wait its Auth Params

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // Set up the ciphering ready for when the UE hands over

    f_UTRAN_ConfigureCipheringBeforeHO_SRVCC(utran_Cell5);

    f_UTRAN_IP_Handling_Change_Configuration (utran_CellDedicated,  cell_DCH_64kPS_RAB_SRB); // @sic R5s140572 sic@

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell SS to transmit a MobilityFromEUTRACommand message

    // @siclog "Steps 32 to 37" siclog@

    //according to 36.508 cl. 6.4.3.9.5,

    f_UTRAN_SRVCC_WithHandover(utran_Cell5, cs_PDP_ContextStatusNSAPI5, true, '0'B); // @sic R5-144670, R5-150398 sic@

    v_DefaultRef := activate (a_UTRAN_DeactivatePDPContext_MO());  

    f_Delay(2.0);
    f_UT_AnswerCall ( UT );

    v_TI_R.tiVal  := '000'B;

    v_TI_R.tiFlag := '1'B; //@sic R5s140578 sic@

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_Connect(v_TI_R))); 

    //@siclog "Step 38" siclog@

    v_TI_S := v_TI_R;

    v_TI_S.tiFlag := '0'B;//@sic R5s140578 sic@

    //@siclog "Step 39" siclog@

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_U_ConnectAck(v_TI_S)));

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // new optional deregistration waits for a trigger @sic R5-154047 sic@

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // @sic R5-144670 sic@

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);

    deactivate (v_DefaultRef);
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell to EUTRA PTC Test case is now finished on E-UTRA side

    f_UTRAN_TestBody_Set(false);

    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U5_CSCALL, -, v_TI_S);

  }
Change 7
	Function name
	f_TC_13_4_3_16_IMS1

	Reason for change
	Removed the trigger to UTRAN as it is not neede anymore as PDP context deativation is moved to default behaviour. (Altstep dismantled).

	Summary of change
	Removed trigger to UTRAN PTC.

	TTCN module
	\LTE_A_IRAT\13\ASRVCC_EUTRA_IMS.ttcn

	MCC160 Comment
	


Before change

function f_TC_13_4_3_16_IMS1 ()runs on IMS_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MT call

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp; /* @sic R5w140112: v_InviteRequestWithSdp instead of v_InviteRequest sic@ */

    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;

    var boolean v_DeregisteredEarly := false;

    f_IMS_PTC_Init();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);  // @sic R5s140587 sic@

    f_IMS_Registration();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);  // @sic R5s140587 sic@

    // Steps 1 to 22 of the generic test procedure for IMS MT speech call (TS 36.508 4.5A.7.3-1).

    /* Where TS 36.508 4.5A.7.3-1 Steps 16 to 22 specifies:

       Steps 5-11 expected sequence defined in annex C.11 of TS 34.229-1 [35]. MTSI MT speech.

    */

    //Wait for Co-ordination message from EUTRA

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_AnnexC11_INVITE();

    v_ResponseWithSdp := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequestWithSdp);

    // Inform EUTRA: IMS Call established.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    //Wait for Co-ordination message from EUTRA

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    f_IMS_MTCallSetup_AnnexC11_Steps5_11(v_InviteRequestWithSdp, v_ResponseWithSdp);

    // Inform EUTRA: IMS Call established.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_PTC_SetRanType(UTRAN_FDD);

    v_DeregisteredEarly := f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5-154047 sic@

    if (not v_DeregisteredEarly) {  // tell UTRAN to stop waiting for Deactivate PDP context @sic R5-162791 sic@

      f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN, cms_IMS_IPCAN_Abort);

    }
    f_IMS_IPCAN_WaitForTrigger(OtherIPCAN);  // @sic R5s140835 sic@
    if (not v_DeregisteredEarly) {  // @sic R5-154047 sic@

      f_IMS_SRVCC_RemoveEarlyDialogMT_AnnexC35(v_InviteRequestWithSdp); // @sic R5-144670 sic@

      f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

      v_DeregisteredEarly := f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5-142919 sic@

    } else {

      f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);  // Still need to tell UTRAN to close down

      f_IMS_IPCAN_WaitForTrigger(OtherIPCAN); // Already done deregistration, so trigger sent during UTRAN postamble needs to be swallowed here

    }

  }
After change

  function f_TC_13_4_3_16_IMS1 ()runs on IMS_PTC

  { // Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / aSRVCC / MT call

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp; /* @sic R5w140112: v_InviteRequestWithSdp instead of v_InviteRequest sic@ */

    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;

    var boolean v_DeregisteredEarly := false;

    f_IMS_PTC_Init();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);  // @sic R5s140587 sic@

    f_IMS_Registration();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);  // @sic R5s140587 sic@

    // Steps 1 to 22 of the generic test procedure for IMS MT speech call (TS 36.508 4.5A.7.3-1).

    /* Where TS 36.508 4.5A.7.3-1 Steps 16 to 22 specifies:

       Steps 5-11 expected sequence defined in annex C.11 of TS 34.229-1 [35]. MTSI MT speech.

    */

    //Wait for Co-ordination message from EUTRA

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_AnnexC11_INVITE();

    v_ResponseWithSdp := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequestWithSdp);

    // Inform EUTRA: IMS Call established.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    //Wait for Co-ordination message from EUTRA

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    f_IMS_MTCallSetup_AnnexC11_Steps5_11(v_InviteRequestWithSdp, v_ResponseWithSdp);

    // Inform EUTRA: IMS Call established.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_PTC_SetRanType(UTRAN_FDD);
    v_DeregisteredEarly := f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5-154047 sic@

    f_IMS_IPCAN_WaitForTrigger(OtherIPCAN);  // @sic R5s140835 sic@

    if (not v_DeregisteredEarly) {  // @sic R5-154047 sic@

      f_IMS_SRVCC_RemoveEarlyDialogMT_AnnexC35(v_InviteRequestWithSdp); // @sic R5-144670 sic@

      f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

      v_DeregisteredEarly := f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5-142919 sic@

    } else {

      f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);  // Still need to tell UTRAN to close down

      f_IMS_IPCAN_WaitForTrigger(OtherIPCAN); // Already done deregistration, so trigger sent during UTRAN postamble needs to be swallowed here

    }

  }
Change 8
	Function name
	f_TC_13_4_3_18_UTRAN

	Reason for change
	Added delay of 2 second before sending Alerting message to provide UE with enough time to proceed with optional deregistration after RAU complete message is received..

Activated default to process the optional PDP context deactivation request/accept after RAU procedure.

	Summary of change
	· Added 2s delay.

· Activated default behaviour.

· Dismantled altstep

	TTCN module
	\LTE_A_IRAT\13\BSRVCC_EUTRA_UTRAN.ttcn

	MCC160 Comment
	


Before change

  function f_TC_13_4_3_18_UTRAN() runs on UTRAN_IRAT_PTC

  { /* @desc  Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / bSRVCC / MO call */

    var integer v_T0_CPICH_Cell5;

    var integer v_T1_CPICH_Cell5;

    var integer v_T0_PCCPCH_Cell5;

    var integer v_T1_PCCPCH_Cell5;

    var TI v_TI_R;

    var TI v_TI_S;

    v_T0_CPICH_Cell5 := -88;

    v_T1_CPICH_Cell5 := -64;

    v_T0_PCCPCH_Cell5 := -88;

    v_T1_PCCPCH_Cell5 := -64;

    // Initialise all cells

    f_UTRAN_Init(EUTRA_UTRAN);

    // Create and configure Cell 5

    // Cell started switched on with default power level, -60 and max attenuation

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    // Start sending System Information

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    // Send relevant UTRAN cell info to EUTRA PTC

    f_UTRAN_Send_IR_SysInfoToEUTRA(utran_Cell5);

    f_UTRAN_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    f_SS_Config_ConvSpeech_64kPS_RAB(utran_Cell5);

    // wait for trigger from EUTRAN side

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    // Set Cell Power level as per T0

    f_UTRAN_SetCellPower(utran_Cell5, v_T0_CPICH_Cell5, v_T0_PCCPCH_Cell5);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 18" siclog@

    // Set Cell Power level as per T1

    f_UTRAN_SetCellPower(utran_Cell5, v_T1_CPICH_Cell5, v_T1_PCCPCH_Cell5);

    // UE starts registered in EUTRA, wait its Auth Params

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // Set up the ciphering ready for when the UE hands over

    f_UTRAN_ConfigureCipheringBeforeHO_SRVCC(utran_Cell5);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell SS to transmit a MobilityFromEUTRACommand message

    f_IRAT_WaitForCoOrd_Trigger(EUTRA); // @sic R5s160330 sic@

    f_UTRAN_IP_Handling_Change_Configuration(utran_CellDedicated, cell_DCH_64kPS_RAB_SRB); // @sic R5s140572 sic@

    //@siclog "Step 23-33" siclog@

    f_UTRAN_SRVCC_WithHandover(utran_Cell5, cs_PDP_ContextStatusNSAPI5, true, '0'B); //@sic R5s140733 R5-150399 sic@

    v_TI_S := {tiFlag := '0'B, tiVal := '000'B};

    v_TI_R := {tiFlag := '1'B, tiVal := '000'B};

    //@siclog "Step 34" siclog@

    // The SS transmits an Alerting message

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                             tsc_RB3,

                             cs_U_Alert(v_TI_S)));

    //@siclog "Step 35" siclog@

    // The SS transmits a Connect message

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                             tsc_RB3,

                             cs_U_Connect(v_TI_S)));

    alt { // @sic R5-162791 sic@
      //@siclog "Step 36" siclog@

      //Receive Connect Acknowledge message

    []  U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,

                                          tsc_RB3,

                                          cr_U_ConnAck(v_TI_R))) {

      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 36");

      }

    [] a_UTRAN_DeactivatePDPContext_MO () {

        repeat;

      }

    }
    f_UTRAN_TestBody_Set(false);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell to EUTRA PTC Test case is now finished on E-UTRA side

    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U5_CSCALL, -, v_TI_S);

  } // f_TC_13_4_3_18_UTRAN
After change

function f_TC_13_4_3_18_UTRAN() runs on UTRAN_IRAT_PTC

  { /* @desc  Inter-system mobility / E-UTRA PS voice + PS data to UTRA CS voice + PS data / bSRVCC / MO call */

    var integer v_T0_CPICH_Cell5;

    var integer v_T1_CPICH_Cell5;

    var integer v_T0_PCCPCH_Cell5;

    var integer v_T1_PCCPCH_Cell5;

    var TI v_TI_R;

    var TI v_TI_S;

    var default v_DefaultRef;
    v_T0_CPICH_Cell5 := -88;

    v_T1_CPICH_Cell5 := -64;

    v_T0_PCCPCH_Cell5 := -88;

    v_T1_PCCPCH_Cell5 := -64;

    // Initialise all cells

    f_UTRAN_Init(EUTRA_UTRAN);

    // Create and configure Cell 5

    // Cell started switched on with default power level, -60 and max attenuation

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    // Start sending System Information

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    // Send relevant UTRAN cell info to EUTRA PTC

    f_UTRAN_Send_IR_SysInfoToEUTRA(utran_Cell5);

    f_UTRAN_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    f_SS_Config_ConvSpeech_64kPS_RAB(utran_Cell5);

    // wait for trigger from EUTRAN side

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    // Set Cell Power level as per T0

    f_UTRAN_SetCellPower(utran_Cell5, v_T0_CPICH_Cell5, v_T0_PCCPCH_Cell5);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 18" siclog@

    // Set Cell Power level as per T1

    f_UTRAN_SetCellPower(utran_Cell5, v_T1_CPICH_Cell5, v_T1_PCCPCH_Cell5);

    // UE starts registered in EUTRA, wait its Auth Params

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // Set up the ciphering ready for when the UE hands over

    f_UTRAN_ConfigureCipheringBeforeHO_SRVCC(utran_Cell5);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell SS to transmit a MobilityFromEUTRACommand message

    f_IRAT_WaitForCoOrd_Trigger(EUTRA); // @sic R5s160330 sic@

    f_UTRAN_IP_Handling_Change_Configuration(utran_CellDedicated, cell_DCH_64kPS_RAB_SRB); // @sic R5s140572 sic@

    //@siclog "Step 23-33" siclog@

    f_UTRAN_SRVCC_WithHandover(utran_Cell5, cs_PDP_ContextStatusNSAPI5, true, '0'B); //@sic R5s140733 R5-150399 sic@

    v_DefaultRef := activate (a_UTRAN_DeactivatePDPContext_MO());

    f_Delay(2.0);
    v_TI_S := {tiFlag := '0'B, tiVal := '000'B};

    v_TI_R := {tiFlag := '1'B, tiVal := '000'B};

    //@siclog "Step 34" siclog@

    // The SS transmits an Alerting message

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                             tsc_RB3,

                             cs_U_Alert(v_TI_S)));

    //@siclog "Step 35" siclog@

    // The SS transmits a Connect message

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                             tsc_RB3,

                             cs_U_Connect(v_TI_S)));

    //@siclog "Step 36" siclog@

    //Receive Connect Acknowledge message

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,

                                          tsc_RB3,

                                          cr_U_ConnAck(v_TI_R))); 

    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 36");

    f_UTRAN_TestBody_Set(false);

    deactivate (v_DefaultRef);
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell to EUTRA PTC Test case is now finished on E-UTRA side

    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U5_CSCALL, -, v_TI_S);

  } // f_TC_13_4_3_18_UTRAN

